
Introduction
This article will describe the steps necessary to deploy Windows-based images on workstations, using TeraByte Image products: Image for DOS, Image for Linux, and Image for Windows.
Data Storage Size Unit Conventions

Since this document will be making reference to data storage size units, it is necessary to provide some clarification on the definitions being used.  Storage device manufacturers typically define gigabytes (GB) in base decimal, where 1 GB = 1,000 MB = 1,000,000,000 bytes.  Microsoft Windows, on the other hand, defines GB in base binary, where 1 GB = 1,024 MB = 1,073,741,824 bytes.
Because of the confusion that can result when these different data storage size unit conventions are each referred to as “gigabytes”, the gibibyte (along with the kibibyte, mebibyte, etc.) was established in 1998 by the International Electrotechnical Commission (IEC).  A gibibyte (abbreviated GiB) is a base binary unit, so 1 GiB = 230 bytes = 1,073,741,824 bytes.  The IEC retained the term gigabyte to refer to base decimal, where 1 GB = 109 bytes = 1,000,000,000 bytes.
This document will follow IEC recommendations, and will thus use the term gigabyte (GB) to refer to base decimal, and gibibytes (abbreviated GiB) to refer to base binary.  In other words, when reference is made to the data storage size convention used by Windows, the units will be shown as gibibytes (GiB), even though Windows itself refers to the units as gigabytes (GB).
Outline of Deploying with TeraByte Image Products
The basic steps involved in deploying with TeraByte Image products are as follows:

I. Establish the source partition(s).

II. Run the Sysprep tool on the source Windows partition.
III. Use a TeraByte product to create an image of the source partition(s).

IV. Create a restore script, and run it to invoke the desired Image product on the destination workstations.

These steps will be elaborated on in the sections to follow.

I. Establish the Source Partition(s)
1. Install Windows as you normally would.  The partition used should be as small as reasonably possible, or at least as small as will be needed on any of the destination workstations.  When the image of the source partition is restored to target workstations, you will be able to expand it to a larger size, if needed.  Deploying the image to partition that is smaller than the source partition is generally not nearly as easy.
2. Install any additional software you want present on the destination systems.
3. Configure Windows and the installed software, as needed.
Example 1: 100-GB Source HD; 50-GB Target HD
The source hard drive is rated at 100 GB (93.1 GiB) in size.  However, since some of the target systems have hard drives with rated capacities as small as 50 GB (46.6 GiB), the source Windows partition can be sized no larger than about 46 GiB (to ensure that the partition is small enough to fit on 46.6-GiB target hard drives when the restore is performed).

If needed, the TeraByte Unlimited utility BootIt NG can be used to resize the source partition prior to copying it.

Please refer to Restore Example 1 for an example of how this configuration could be restored via script, with the target systems being booted locally, and the image files residing on CD/DVD discs.  A bootable floppy diskette image is provided for this example.
Example 2: 80-GB Source HD; 60-GB Target HD (40 GiB/15.9 GiB)
The source hard drive is rated at 100 GB (93.1 GiB) in size, and is divided into two partitions: One larger partition for Windows and applications, and one smaller partition for data.  The smallest target hard drive is 60 GB (55.9 GiB) in size, and on these systems, the Windows partition will be made 40 GiB in size, and the data partition will be 15.9 GiB in size.

In order to facilitate a restore to target hard drives with 40 GiB/15.9 GiB partitions, the source hard drive is set up with a 30-GiB Windows partition, and a 10-GiB data partition.  It isn’t necessary to make the source partitions as small as 30 GiB/10 GiB (the sizes need only be somewhat smaller than 40 GiB/15.9 GiB), but since those sizes were still large enough to accommodate the data to be imaged, they were chosen for efficiency.

If needed, the TeraByte Unlimited utility BootIt NG can be used to resize the source partitions prior to copying them.

Please refer to Restore Example 2 for an example of how this configuration could be deployed via script, with the target systems being booted locally, and the image files residing on CD/DVD discs.  A bootable floppy diskette image is provided for this example.
Please refer to Restore Example 3 for an example of how this configuration could be restored via script, with the target systems being booted locally, and the image files residing on a mapped network drive.
Please refer to Restore Example 4 for an example of how this configuration could be restored via script, with the target systems being booted via Linux PXE, and the image files residing on a mapped network drive.

II. Run the Sysprep Tool on the Source Windows Partition
Accommodating Mass Storage Controller Types
To accommodate target system mass storage controllers that differ from that of the source system, you have two options:

1. In the Sysprep answer file (i.e. sysprep.inf), include the BuildMassStorageSection=Yes setting in the [Sysprep] section, and also include an empty [SysprepMassStorage] section.  For example:

[Sysprep]

BuildMassStorageSection=Yes
[SysprepMassStorage]

If you choose this option, the [SysprepMassStorage] section will be populated when Sysprep is run.


2. Exclude the BuildMassStorageSection setting, and include only an empty [SysprepMassStorage] section.  For example:

[SysprepMassStorage]

If you choose this option, it will be necessary to run the command sysprep -bmsd separately, prior to running Sysprep normally.  The advantage to this method (as opposed to the method mentioned in number 1 above) is that you will have the opportunity to manually remove unnecessary mass storage controller driver references from the [SysprepMassStorage] section.  The sysprep -bmsd command will be mentioned in the Running Sysprep on the Source System section to follow.
Accommodating Uniprocessor and Multiprocessor HAL Types

If one or more of the target systems uses a hardware abstraction layer (HAL) that differs from that of the source system, you will need to account for this in a Sysprep answer file (i.e. sysprep.inf).  For example, if the source system uses the multiprocessor HAL, one or more of the target systems will use the uniprocessor HAL, and if both systems use ACPI, place the following lines in sysprep.inf:

[Unattended]
UpdateUPHAL = "ACPIAPIC_UP,%WINDIR%\Inf\Hal.inf"
If the source system uses the multiprocessor HAL, one or more of the target systems will use the uniprocessor HAL, and both systems do not use ACPI, place the following lines in sysprep.inf:

[Unattended]
UpdateUPHAL = "MPS_UP,%WINDIR%\Inf\Hal.inf"

If the source system uses the uniprocessor HAL, one or more of the target systems will use the multiprocessor HAL, and both systems use ACPI, place the following lines in sysprep.inf:

[Unattended]
UpdateHAL = "ACPIAPIC_MP,%WINDIR%\Inf\Hal.inf""

Please note that in this case, the multiprocessor HAL will always be in effect on target systems, which may not be desired.  For this reason, it may be a good idea to use a multiprocessor system as the source, and allow uniprocessor HALs to be applied as necessary.
If you use a Sysprep answer file, sysprep.inf, place it in the Sysprep directory as explained in step 3 in the section titled Running Sysprep on the Source System immediately below.
Running Sysprep on the Source System
1. Open a command window in the source Windows installation, and then issue the following commands:

cd c:\
md sysprep
cd c:\sysprep

This will create the folder C:\sysprep, and make it the current directory in the command window.
2. Locate the appropriate copy of the file deploy.cab.  The copy of deploy.cab you use should be the one that was distributed by Microsoft specifically for the Windows service pack you are using (if any).  If you are using a standard or slipstreamed Windows CD-ROM, you should be able to find the correct copy of deploy.cab on the CD-ROM under the folder \support\tools\.  If you installed a service pack after installing Windows, you should be able to locate the appropriate copy of deploy.cab by searching the Microsoft Download Center (e.g. search for “Windows 2000 Service Pack 4 deployment tools” or “Windows XP Service Pack 2 deployment tools”).
3. Extract the files sysprep.exe and setupcl.exe from deploy.cab, and place them in C:\sysprep (the folder created in step 1).  If you are using a sysprep.inf answer file, that file will also need to be copied to C:\sysprep.
4. If (and only if) you omitted the BuildMassStorageSection setting (as mentioned in the Accommodating Mass Storage Controller Types section above), run the command sysprep –bmsd in the command window opened in step 1.  This will populate The [SysprepMassStorage] section of sysprep.inf.  If desired, and after running the sysprep –bmsd command, you may manually delete entries that you know will not be needed from the [SysprepMassStorage] section.  Deleting unnecessary entries may speed up the initial operating system boot process.
5. Run Sysprep in the command window opened in step 1 by running the command sysprep.  The System Preparation Tool window will appear.  You may also choose to run sysprep with command line parameters.  Please refer to the applicable documentation for more information (such as the help file named deploy.chm or deptool.chm, which is distributed in deploy.cab.
III. Use a TeraByte Product to Create an Image of the Source Partition(s)
All TeraByte products that are capable of creating images—Image for DOS, Image for Linux, Image for Windows, and BootIt NG—produce image files that are fully compatible with one another, so long as the major version number of the products being used is the same.  That is, an image created with a version-1.x TeraByte product can only be later restored with a version-1.x TeraByte product.  Likewise, an image created with a version-2.x TeraByte product* can only be later restored with a version-2.x TeraByte product.  The products used to image and restore need not be the same; they only need to be of the same major version level.
By default, all TeraByte imaging products copy only used sectors, and apply maximum compression to the image set.  Therefore, you can generally create the image of the source partition using default settings.
Where image files are to be stored will depend on the nature and extent of your overall deployment scenario.  For example, small and/or specialized deployments may be able to store image files on DVD discs or an external USB hard drive, while larger and/or more extensive deployments often require the use of network drives for the storing of image files.  The restore examples provided in this document include both types of scenarios.
* At the time of this writing, version-2.x TeraByte product releases are forthcoming.

IV. Create and Run a Restore Script

A restore script can be used to invoke Image for DOS, Image for Linux, or Image for Windows.  Such a script could be run from a bootable restore floppy diskette or CD/DVD disc, to semi-automate the deployment process.  Or, if needed, you could place the script (and the Image product executable) on a network drive, and run the script from there.  If you will be restoring from a network drive, you have several choices, including (but not limited to):

· Use Image for Windows and point it to a UNC path or a mapped network drive to restore from.

· Use a network-enabled (or PXE) Image for DOS boot diskette or disc.

· Use a network-enabled (or PXE) Image for Linux boot disc.  A trial version of a PXE-enabled Image for Linux boot disc is available here, and a registered version may be obtained here.
Below, copies of the applicable example scripts are shown in order to include syntax highlighting, to enhance readability.  For working copies of the first two scripts, please find the example1.ima and example2.ima files included in the archive used to package this document.  (If necessary, the archive can be obtained at this link.)

Restore Example 1

This example involves the use of a batch file that is run from a bootable FreeDOS diskette:
@echo off

:: This is a sample deployment script that corresponds to "Example 1" in the

:: "Deploying with TeraByte Image Utilities" guide.  This script is designed

:: to run under FreeDOS, using Image for DOS.  This choice of environment and

:: application is for illustrative purposes only; in actual use, you are not

:: restricted to using the environment and/or TeraByte application used here.

cls

cd \

:: Running CADEL.COM prevents Ctrl+Alt+Delete, Ctrl+C, and Ctrl+Break from

:: working, to help prevent unintended direct access to the FreeDOS medium.

::

:: CADEL.COM is freeware, available from

:: ftp://ftp.simtel.net/pub/simtelnet/msdos/keyboard/cadel.zip

\utils\cadel.com

cls

echo.

echo Please wait...

echo.

:: Checking to see whether the script is being run under FreeDOS:

::

:: FIND.COM is freeware, available from

:: http://www.coli.uni-saarland.de/~eric/stuff/soft/

ver | \utils\find.com "FreeCom" >nul

if not ErrorLevel 1 goto :Skip01

cls

echo.

echo %0:

echo.

echo This script should be run only under FreeDOS.

echo.

pause

goto :NoAction

:Skip01

:: Since image.exe is subject to replacement (to replace it with a registered

:: copy, upgrade it, etc.), it may be a good idea to check for its existence.
:: Note: The "set OS=FD" command must not precede this section of code:
\image.exe >nul
if ErrorLevel 1 goto :Skip02

cls

echo.

echo The file IMAGE.EXE was not found.

echo.

pause

goto :NoAction

:Skip02

:: The showjpg.exe line below results in the image file \utils\hd.jpg being

:: displayed for 4,000 ms (4 seconds).

::

:: Showjpg.exe is freeware, available from http://www.pictview.com

\utils\showjpg.exe \utils\hd.jpg -t4000

:: Running MSGBOX.EXE allows customized warnings/prompts to be displayed.

:: The working directory is changed in order to point to specific copies

:: of MSGBOX.TXT, which is where MSGBOX.EXE reads the text to be displayed.

::

:: MSGBOX.EXE is freeware, available from

:: http://www.terabyteunlimited.com/utilities.html#msgbox

:: First, an initial restore warning message is displayed.  The message shown

:: on screen originates from \utils\msgbox01\msgbox.txt:

cd \utils\msgbox01\

..\msgbox.exe yn
if ErrorLevel 2 goto :NoAction

:: If the user tells the first MSGBOX.EXE to continue, a second, final warning

:: displayed.  This message text originates from \utils\msgbox02\msgbox.txt:

cd \utils\msgbox02\

..\msgbox.exe 10
if ErrorLevel 2 goto :NoAction

if ErrorLevel 1 goto :Restore

:NoAction

cls

echo.

echo No changes have been made to the system.

echo.

echo The system will now reboot.

echo.

pause

goto :Reboot

:Restore

:: The SET command below is necessary to prevent the error "This program must

:: be run under DOS and not from a DOS Box" from occurring when running IFD.

set OS=FD

:: Breakdown of image.exe command line found below (/R0NTAX /CR /CD#IMG01):

::    /R0 ......... Restore to HD 0, to partition ID matching source
::    N ........... Suppress the post-restore reboot prompt
::    T ........... Restore standard MBR

::    A ........... Set restored partition active

::    X ........... Expand restored partition to fill any leftover space

::    /CR ......... Clear MBR and EMBR, and restore NT disk signature

::    /CD#IMG01 ... Restore file IMG01 from first CD/DVD drive found on system

cd \

\image.exe /R0NTAX /CR /CD#IMG01

set ErrorCode=0

if ErrorLevel 1 set ErrorCode=%ErrorLevel%

echo.

if     "%ErrorCode%"=="0" echo The restore completed without error.

if not "%ErrorCode%"=="0" echo The restore ended with error %ErrorCode%.

echo.

echo The system will now reboot.

pause

goto :Reboot

:Reboot

:: Running MBR.EXE with the "0 /reboot" parameters simply performs a reboot.

:: The "mbr.exe 0 /reboot" command does not make any changes to the disk.

::

:: MBR.EXE is freeware, available from

:: http://www.terabyteunlimited.com/utilities.html#mbr

\utils\mbr.exe 0 /reboot

:: End of script

Restore Example 2

This example involves the use of a batch file that is run from a bootable FreeDOS diskette:

@echo off

:: This is a sample deployment script that corresponds to "Example 2" in the

:: "Deploying with TeraByte Image Utilities" guide.  This script is designed

:: to run under FreeDOS, using Image for DOS.  This choice of environment and

:: application is for illustrative purposes only; in actual use, you are not

:: restricted to using the environment and/or TeraByte application used here.

cls

cd \

:: Running CADEL.COM prevents Ctrl+Alt+Delete, Ctrl+C, and Ctrl+Break from

:: working, to help prevent unintended direct access to the FreeDOS medium.

::

:: CADEL.COM is freeware, available from

:: ftp://ftp.simtel.net/pub/simtelnet/msdos/keyboard/cadel.zip

\utils\cadel.com

cls

echo.

echo Please wait...

echo.

:: Checking to see whether the script is being run under FreeDOS:

::

:: FIND.COM is freeware, available from

:: http://www.coli.uni-saarland.de/~eric/stuff/soft/

ver | \utils\find.com "FreeCom" >nul

if not ErrorLevel 1 goto :Skip01

cls

echo.

echo %0:

echo.

echo This script should be run only under FreeDOS.

echo.

pause

goto :NoAction

:Skip01

:: Since image.exe is subject to replacement (to replace it with a registered

:: copy, upgrade it, etc.), it may be a good idea to check for its existence.
:: Note: The "set OS=FD" command must not precede this section of code:
\image.exe >nul
if ErrorLevel 1 goto :Skip02

cls

echo.

echo The file IMAGE.EXE was not found.

echo.

pause

goto :NoAction

:Skip02

:: The showjpg.exe line below results in the image file \utils\hd.jpg being

:: displayed for 4,000 ms (4 seconds).

::

:: Showjpg.exe is freeware, available from http://www.pictview.com

\utils\showjpg.exe \utils\hd.jpg -t4000

:: Running MSGBOX.EXE allows customized warnings/prompts to be displayed.

:: The working directory is changed in order to point to specific copies

:: of MSGBOX.TXT, which is where MSGBOX.EXE reads the text to be displayed.

::

:: MSGBOX.EXE is freeware, available from

:: http://www.terabyteunlimited.com/utilities.html#msgbox

:: First, an initial restore warning message is displayed.  The message shown

:: on screen originates from \utils\msgbox01\msgbox.txt:

cd \utils\msgbox01\

..\msgbox.exe yn
if ErrorLevel 2 goto :NoAction

:: If the user tells the first MSGBOX.EXE to continue, a second, final warning

:: displayed.  This message text originates from \utils\msgbox02\msgbox.txt:

cd \utils\msgbox02\

..\msgbox.exe 10
if ErrorLevel 2 goto :NoAction

if ErrorLevel 1 goto :Restore

:NoAction

cls

echo.

echo No changes have been made to the system.

echo.

echo The system will now reboot.

echo.

pause

goto :Reboot

:Restore

:: The SET command below is necessary to prevent the error "This program must

:: be run under DOS and not from a DOS Box" from occurring when running IFD.

set OS=FD

:: Breakdown of image.exe command line found below (/R0NTAX16268 /CR /CD#IMG01):

::    /R0 ......... Restore to HD 0, to partition ID matching source
::    N ........... Suppress the post-restore reboot prompt

::    T ........... Restore standard MBR

::    A ........... Set restored partition active

::    X16268 ...... Expand restored partition to fill leftover space,

::                  minus 16268 MB, which will be needed for data volume
::    /CR ......... Clear MBR and EMBR, and restore NT disk signature

::    /CD#IMG01 ... Restore file IMG01 from first CD/DVD drive found on system
cd \

\image.exe /R0NTAX16268 /CR /CD#IMG01
set ErrorCode=0

if ErrorLevel 1 set ErrorCode=%ErrorLevel%

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The restore of IMG01 ended with error %ErrorCode%.

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The system will now reboot.

if not "%ErrorCode%"=="0" pause

if not "%ErrorCode%"=="0" goto :Reboot
:: Breakdown of image.exe command line found below (/R02NXM /CD#IMG02):

::    /R02 ........ Restore to HD 0, partition 2

::    N ........... Suppress the post-restore reboot prompt

::    X ........... Expand restored partition to fill any leftover space
::    M ........... Restore to first block of free space that is large enough
::    /CD#IMG02 ... Restore file IMG02 from first CD/DVD drive found on system
\image.exe /R02NXM /CD#IMG02

set ErrorCode=0

if ErrorLevel 1 set ErrorCode=%ErrorLevel%

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The restore of IMG02 ended with error %ErrorCode%.

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The system will now reboot.

if not "%ErrorCode%"=="0" pause

if not "%ErrorCode%"=="0" goto :Reboot

:Reboot

:: Running MBR.EXE with the "0 /reboot" parameters simply performs a reboot.

:: The "mbr.exe 0 /reboot" command does not make any changes to the disk.

::

:: MBR.EXE is freeware, available from

:: http://www.terabyteunlimited.com/utilities.html#mbr

\utils\mbr.exe 0 /reboot

:: End of script

Restore Example 3

This example involves the use of a batch file that is run from a mapped network drive.  Although the batch file runs from a network drive, the target systems in this example are still booted locally (please see Restore Example 4 if you are interested in booting directly from the network).  To run the batch file, a network-enabled bootable diskette such as Bart’s Network Boot Disk or a NetWare Boot Disk is used.  So, assuming that the following conditions are true:

· Source partitions have been established.
· Sysprep has been run on the source Windows partition.
· A TeraByte Image utility has been used to create images of the two partitions.
The basic steps that need to be carried out to implement this restore scenario are as follows:
1. Place the image files on the desired network drive.  In this example, image files will be placed in the network directory \\server01\data\, which will be mapped to drive letter R:.
2. Place the batch file and the TeraByte Image utility executable of choice (e.g. the Image for DOS executable image.exe) on the desired network drive.  In this example, image files will be placed in the network directory \\server02\deploy\, which will be mapped to drive letter S:.  The directory used could 
3. Boot with a network-enabled bootable DOS diskette, such as Bart’s Network Boot Disk.
4. Map the applicable network drives with commands such as:

net use r: \\server01\data\
net use s: \\server02\deploy\

These commands could be put in an autoexec.bat file on the bootable diskette.
5. Run the script to initiate the restore:
@echo off

:: This is a sample deployment script that corresponds to "Example 3" in the

:: "Deploying with TeraByte Image Utilities" guide.  This script is designed

:: to run under MS-DOS, using Image for DOS.  This choice of environment and

:: application is for illustrative purposes only; in actual use, you are not

:: restricted to using the environment and/or TeraByte application used here.

cls

cd \

:: Running CADEL.COM prevents Ctrl+Alt+Delete, Ctrl+C, and Ctrl+Break from

:: working, to help prevent unintended direct access to the FreeDOS medium.

::

:: CADEL.COM is freeware, available from

:: ftp://ftp.simtel.net/pub/simtelnet/msdos/keyboard/cadel.zip

\utils\cadel.com

cls

echo.

echo Please wait...

echo.

:Skip01

:: Since image.exe is subject to replacement (to replace it with a registered

:: copy, upgrade it, etc.), it may be a good idea to check for its existence.

S:\image.exe >nul
if ErrorLevel 1 goto :Skip02

cls

echo.

echo The file S:\IMAGE.EXE was not found.

echo.

pause

goto :NoAction

:Skip02

:: The showjpg.exe line below results in the image file \utils\hd.jpg being

:: displayed for 4,000 ms (4 seconds).

::

:: Showjpg.exe is freeware, available from http://www.pictview.com

\utils\showjpg.exe \utils\hd.jpg -t4000
:: Running MSGBOX.EXE allows customized warnings/prompts to be displayed.

:: The working directory is changed in order to point to specific copies

:: of MSGBOX.TXT, which is where MSGBOX.EXE reads the text to be displayed.

::

:: MSGBOX.EXE is freeware, available from

:: http://www.terabyteunlimited.com/utilities.html#msgbox

:: First, an initial restore warning message is displayed.  The message shown

:: on screen originates from \utils\msgbox01\msgbox.txt:

cd \utils\msgbox01\

..\msgbox.exe yn
if ErrorLevel 2 goto :NoAction

:: If the user tells the first MSGBOX.EXE to continue, a second, final warning

:: displayed.  This message text originates from \utils\msgbox02\msgbox.txt:

cd \utils\msgbox02\

..\msgbox.exe 10
if ErrorLevel 2 goto :NoAction

if ErrorLevel 1 goto :Restore

:NoAction

cls

echo.

echo No changes have been made to the system.

echo.

echo The system will now reboot.

echo.

pause

goto :Reboot

:Restore

:: Breakdown of image.exe command line found below (/R0NTAX16268 /CR /R:\IMG01):

::    /R0 ......... Restore to HD 0, to partition ID matching source
::    N ........... Suppress the post-restore reboot prompt

::    T ........... Restore standard MBR

::    A ........... Set restored partition active

::    X16268 ...... Expand restored partition to fill leftover space,

::                  minus 16268 MB, which will be needed for data volume
::    /CR ......... Clear MBR and EMBR, and restore NT disk signature

::    R:\IMG01 .... Restore file IMG01 from mapped network drive R:
cd \

S:\image.exe /R0NTAX16268 /CR R:\IMG01
set ErrorCode=0

if ErrorLevel 1 set ErrorCode=%ErrorLevel%

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The restore of IMG01 ended with error %ErrorCode%.

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The system will now reboot.

if not "%ErrorCode%"=="0" pause

if not "%ErrorCode%"=="0" goto :Reboot
:: Breakdown of image.exe command line found below (/R02NXM R:\IMG02):

::    /R02 ........ Restore to HD 0, partition 2

::    N ........... Suppress the post-restore reboot prompt

::    X ........... Expand restored partition to fill any leftover space
::    M ........... Restore to first block of free space that is large enough
::    R:\IMG02 .... Restore file IMG02 from mapped network drive R:
S:\image.exe /R02NXM R:\IMG02

set ErrorCode=0

if ErrorLevel 1 set ErrorCode=%ErrorLevel%

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The restore of IMG02 ended with error %ErrorCode%.

if not "%ErrorCode%"=="0" echo.

if not "%ErrorCode%"=="0" echo The system will now reboot.

if not "%ErrorCode%"=="0" pause

if not "%ErrorCode%"=="0" goto :Reboot

:Reboot

:: Running MBR.EXE with the "0 /reboot" parameters simply performs a reboot.

:: The "mbr.exe 0 /reboot" command does not make any changes to the disk.

::

:: MBR.EXE is freeware, available from

:: http://www.terabyteunlimited.com/utilities.html#mbr

\utils\mbr.exe 0 /reboot

:: End of script
Restore Example 4
The scenario for this example is as follows:

· Image files reside on a mapped network drive.

· Target systems boot via Wake-On-LAN (WOL) via a Linux PXE server.

· Image for Linux is used to perform the restore, via Linux script.
The requirements for deploying using this method are as follows:

· Each target workstation must have:

· A system BIOS capable of booting from the network.

· A system BIOS configured to boot from the network.

· A network adapter compatible with network booting.

· The network must have:

· A TFTP server to send image files to target workstations.

· A TFTP server that supports the tsize option (e.g. Advanced TFTP Daemon or atpftd).

· A DHCP server that points to the TFTP server.

The remainder of this example assumes that the following conditions are true:

· Source partitions have been established.

· Sysprep has been run on the source Windows partition.

· A TeraByte Image utility has been used to create images of the two partitions.

The basic steps that need to be carried out to implement this restore scenario are as follows:

1. Download the Image for Linux Network Boot Disc trial version or licensed version.
2. After extracting the Image for Linux Network Boot Disc archive, refer to quickstart.txt and readme.txt.
3. To use an Image for Linux restore script, place the script in the scripts directory that is created when config.zip is extracted.  As noted in readme.txt, to make this script run upon boot, edit the RUNSCRIPTS option in config.txt.
4. Run makePXE.

5. Refer to pxe.txt for information on performing a network boot.

Last Revised: August 17, 2007

